

Graphisme assisté par ordinateur sur Canon X 07

Le Canon X 07 possède 64 caractères graphiques redéfinissables au gré de l'utilisateur par la fonction FONT\$. Mais le calcul des paramètres de cette fonction devient vite fastidieux. Si vous souhaitez composer un dessin utilisant plusieurs dizaines de « FONT\$ » la tâche devient herculéenne !

Le logiciel proposé ici vous permet de dessiner en toute liberté sur l'écran, de calculer puis de mémoriser automatiquement les caractères graphiques créés, et surtout d'archiver sur cassette le dessin réalisé. En prime, une recopie d'écran sur X-710 (rapide !) vous laisse un souvenir impérissable.

Initialisation

Le programme est lancé par RUN. Si vous êtes plusieurs semaines sans utiliser le logiciel, il est possible de revoir rapidement le rôle des touches utiles.

Après les explications (éventuelles) une séquence de musique aléatoire réjouit vos oreilles pendant quelques secondes.

Une pression sur une touche quelconque permet de passer aux choses sérieuses.

L'entrée des paramètres et le choix des options

Tout d'abord, il est nécessaire d'entrer les dimensions du cadre dans lequel s'inscrira le dessin. Ces dimensions sont exprimées en nombre de caractères alphanumériques ($L \leq 20$ et $H \leq 4$, L et H représentant respectivement le nombre de colonnes et le nombre de lignes).

En outre, l'utilisateur ne disposant que de 64 caractères graphiques pour mémoriser un dessin, le programme refuse tout produit $L \times H > 64$.

Cette limitation est mise à profit pour créer un espace de « dialogue » entre l'utilisateur et le programme. Cet espace est situé soit dans les trois colonnes de droite si $L \leq 16$, soit sur la dernière ligne dans le cas contraire (en effet, la surface-

écran maxi utilisable est soit 16×4 , soit 20×3).

- Le programme laisse ensuite la possibilité de définir le numéro du premier caractère graphique à utiliser pour la mémorisation (dans la limite, bien sûr, des 64 possibles). Cette option est particulièrement intéressante si vous voulez sauvegarder plusieurs petits dessins à la fois (par exemple 4-dessins de 4×4), utilisables ensuite pour une animation graphique.

- Le choix de l'option « sauvegarde sur cassette » vous permet de donner un titre au dessin. Ce titre sera mémorisé en début d'enregistrement.

- Il est possible de sortir sur imprimante les paramètres décimaux des « FONT\$ » calculés (fig. 1). A ce moment, la sortie n'est possible que sur imprimante puisque l'écran est « protégé » pendant la phase de mémorisation. Par contre, à l'issue de cette phase, la sortie des paramètres des FONT\$ (en hexadécimal) est possible sur écran et/ou sur imprimante.

Le dessin proprement dit

La zone graphique apparaît, délimitée sur l'écran par un cadre. Le curseur est situé en (0,0) et clignote. Pour dessiner, l'utilisateur dispose alors des possibilités suivantes :

- Les quatre touches de déplacement du curseur (◀, ▲, ▶, ▼) permettent de dessiner des droites dans les quatre directions, plus les diagonales à 45° par appui simultané sur deux de ces touches.

- Pour effacer, il suffit d'appuyer sur la touche F6 située au milieu du pavé des touches de déplacement. Un bip sonore est émis et une étoile (*) apparaît

UTILITAIRE : « Course aux FONT\$ » d'Alain NOGUES

Le tracé de formes sur un micro-ordinateur peut être fastidieux. Avec cet utilitaire, cela devient un régal.

Langage : Basic

Ordinateur : Canon X 07

```

1 ' *****
2 ' *
3 ' * COURSE AUX FONT$ *
4 ' * sur Canon X-07 *
5 ' *
6 ' * AUTEUR: Alain NOGUES *
7 ' *
8 ' *****
9 '
20 CLEAR 600:DEFSTRA=D:DEFINTE=Z
21 CONSOLE0,4,,1,1
25 CLS:LOCATE6,0:PRINT"BONJOUR !"
30 LOCATE4,1:PRINT"Voulez-vous "
31 LOCATE1,2:INPUT"des explications";D
32 IFLEFT$(D,1)="0"THENGOSUB700
33 '-----PRESENTATION-----
---
35 J=VAL(RIGHT$(TIME$,2))/4
36 FORI=0TOJ:K=15*RDND(1):NEXT:J=1
40 CLS:FORI=1TO18:LOCATEI,0:PRINT"*";
41 N=K+J*10*RDND(1)
42 LOCATEI,3:PRINT"*";BEEP,N,3:J=J
43 NEXT
45 LOCATE0,1:PRINT"* COURSE AUX FONT$ *"
;
46 LOCATE0,2:PRINT"* sur Canon X-07 *"
;
47 GOSUB700
48 '-----ENTREE DES PARAMETRES-----
---
50 CLS:PRINT"Fixez les dimensions du cadr
e du dessin :";
51 INPUT"LONGUEUR L";L
52 INPUT"HAUTEUR H";H
55 IFL>0ANDL<21GOTO60
56 GOSUB72:PRINT"L doit etre compris ent
re 1 et 20 inclus";
57 GOTO67
60 IFH>0ANDH<5GOTO65
61 GOSUB72:PRINT"H doit etre compris ent
re 1 et 4 inclus";
62 GOTO67
65 IFL*H<65GOTO75
66 GOSUB72:PRINT"Le produit de L par H n
e peut excéder 64";
67 GOSUB70:GOTO50
70 IFINKEY$=""THEN70ELSERETURN
72 CLS:LOCATE3,0:PRINT"ATTENTION !":RETU
RN
73 '---CHOIX DE LA ZONE DE "DIALOGUE"---
---
75 IFL<17THENQ=0ELSEQ=1
76 IFQ=1ANDH=3THENQ=2
77 XL=L*6-1:YL=H*8-1:N=L*H

```

Listing du programme Basic.



```

*HODOCK
FONT#( 128 )=(00,00,00,00,00,00,00,00)
FONT#( 129 )=(00,00,00,00,00,00,0C,3C,7C)
FONT#( 130 )=(00,00,0C,3C,FC,FC,FC,FC)
FONT#( 131 )=(00,00,00,00,00,00,0C,FC,FC)
FONT#( 132 )=(00,00,00,00,00,00,0C,00,00)
FONT#( 133 )=(00,00,00,00,00,00,00,00,00)
FONT#( 134 )=(00,00,00,00,00,00,00,00,00)
FONT#( 135 )=(00,00,00,00,00,00,00,00,00)
FONT#( 136 )=(00,00,00,00,00,00,00,00,00)
FONT#( 137 )=(00,00,00,00,00,00,00,00,00)
FONT#( 138 )=(00,00,00,00,00,00,00,00,00)
FONT#( 139 )=(00,00,00,00,00,00,00,00,00)
FONT#( 140 )=(00,00,00,00,00,00,00,00,00)
FONT#( 141 )=(00,00,00,00,00,00,00,00,00)
FONT#( 142 )=(00,00,00,00,00,00,00,00,00)
FONT#( 143 )=(00,00,00,00,00,00,00,00,00)
FONT#( 144 )=(00,00,00,00,00,00,00,00,00)
FONT#( 145 )=(00,00,00,00,00,00,00,00,00)
FONT#( 146 )=(00,00,00,00,00,00,00,00,00)
FONT#( 147 )=(00,00,00,00,00,00,00,00,00)

```

Fig. 1. - Listing des FONT\$ calculées.

dans le coin inférieur droit, signalant le mode « effacement ». Le déplacement du spot efface alors les points rencontrés par ce dernier.

Pour revenir en mode « dessin », il suffit d'une nouvelle pression sur F6. Un nouveau bip est émis et l'étoile disparaît. ● Il existe deux vitesses de déplacement du spot. Au départ, c'est le mode « rapide » qui est sélectionné. Pour passer en mode « lent », pressez la **barre d'espacement**. Un bip est émis et les coordonnées du spot s'affichent en clignotant dans la zone réservée au « dialogue ».

Une nouvelle pression sur la barre autorise le retour en mode rapide. Les coordonnées ne sont plus affichées.

Pour une prise en compte plus rapide de toutes les commandes en mode lent, pressez la touche désirée lorsque les coordonnées « s'effacent » au cours du clignotement.

● Pour positionner le spot au point de coordonnées ($x < L * 6$, $y < H * 8$), pressez la touche P.

$X = ?$ est alors affiché. Il faut entrer x et appuyer sur **RETURN**. Si $x \geq L * 6$, la valeur entrée n'est pas acceptée.

```

78 CONSOLE,,0
79 '--NUMERO PREMIER CARACTERE GRAPHIQUE
--
80 CLS:INPUT"Voulez-vous fixer le numer
o du premier caractere graphique";D
81 IFLEFT$(D,1)<>"0" THENR=128:GOTO95
85 CLS:INPUT"Numero";R
86 IFR>127ANDR<160ORR>223ANDR<256GOTO90
87 GOSUB72:PRINT"Cette valeur R doit et
re >127 et <160 ou >223 et <256";
88 GOSUB70:GOTO85
90 S=-(192-R)*(R<160)-(256-R)*(R>223)
91 IFN<=SGOTO95
92 GOSUB72:PRINT" Numero trop eleve pou
r les dimensions du cadre";
93 GOSUB70:GOTO85
95 N=L*H
96 '-----OPTION SAUVEGARDE SUR CASSETTE--
--
100 CLS:INPUT"Voulez-vous archiver le
dessin sur cassette";D
101 IFLEFT$(D,1)="0" THENE=1ELSEE=0:GOTO1
10
102 CLS:PRINT"Pensez a positionner la ban
de et a regler tonalite et volume a";
103 LOCATE0,3:PRINT"S/10 ; O.K.?" ;
104 GOSUB70
105 CLS:INPUT" Voulez-vous donner un ti
tre a l'enre- gistrement";D
106 IFLEFT$(D,1)="N" THENCT="SANS TITRE":
GOTO110
107 CLS:LINEINPUT" Quel est ce titre ?";
CT
108 '-----OPTION SORTIE PARAMETRES-----
--
110 CLS:PRINT" Voulez-vous editer les FO
NT$ des caracteres graphiques en";
111 LOCATE0,3:INPUT"notation decimale";D

112 IP=0:IFLEFT$(D,1)<>"0" THEN120
113 CLS:PRINT"L'imprimante est en servic
e ? ; couleur selectionnee ? ";
114 LOCATE3,3:PRINT"Alors O.K. ?";
115 IP=1:GOSUB70
116 '*****DEBUT DU MODULE GRAPHIQUE*****
***
120 CLS:LINE(L*6,0)-(L*6,H*8):LINE-(0,H*
8)
125 X=0:Y=0:P=1:F=1:Z=1
126 GOTO140
127 '-----VITESSE LENTE-----
--
130 A=STR$(X):A=RIGHT$(A,LEN(A)-1):GOSUB
270
131 A=STR$(Y):A=RIGHT$(A,LEN(A)-1):GOSUB
272
135 BEEP0,7:BEEP0,1
136 A=" " :GOSUB270:GOSUB272
137 '-----VITESSE RAPIDE-----
--
140 T=STICK(0)+1:ONTGOSUB150,170,172,175
,177,180,182,185,187
141 PRESET(X,Y):IFP=1 THENPSET(X,Y)
142 IFZ=1 THEN140ELSE130
145 '-----SCRUTATION DU CLAVIER-----
--
150 PSET(X,Y):BEEP0,2:BEEP0,1
151 PRESET(X,Y)
152 IFSTRIG(1)GOTO200
153 IFSTRIG(0)GOTO210
155 IFTKEY("D")GOTO215
156 IFTKEY("P")GOTO220
157 IFTKEY("C")GOTO240
158 IFTKEY("L")GOTO250
159 IFTKEY("M")GOTO300

160 IFTKEY("E")GOTO120
161 IFTKEY("A")GOTO350
162 RETURN
165 '---TRAITEMENT DES TOUCHES DU CURSEUR
---
170 Y=Y-1:IFY<0 THENY=0
171 RETURN
172 Y=Y+1:X=X+1:IFY<0 THENY=0
173 IFX>XL THENX=XL
174 RETURN
175 X=X+1:IFX>XL THENX=XL
176 RETURN
177 X=X-1:Y=Y+1:IFX>XL THENX=XL
178 IFY>YL THENY=YL
179 RETURN
180 Y=Y+1:IFY>YL THENY=YL
181 RETURN
182 Y=Y-1:X=X-1:IFY>YL THENY=YL
183 IFX<0 THENX=0
184 RETURN
185 X=X-1:IFX<0 THENX=0
186 RETURN
187 X=X-1:Y=Y-1:IFX<0 THENX=0
188 IFY<0 THENY=0
189 RETURN
195 '-----EFFACEMENT(F6)-----
--
200 P=1-P:BEEP10,4:BEEP0,1
201 IFP=1 THENA=" " ELSEA="*"
202 GOSUB274
203 IFP=0 THENX1=X:Y1=Y:RETURN
204 IFP=1 THENX2=X:Y2=Y:RETURN
206 '-----CHANGEMENT DE VITESSE-----
--
210 Z=1-Z:BEEP5,4:BEEP0,1
211 IFZ=1 THENA=" " :GOSUB270:GOSUB272
212 RETURN
214 '-----TRACE D'UNE DROITE-----
--
215 LINE(X1,Y1)-(X2,Y2):RETURN
217 '-----POSITIONNEMENT DU SPOT-----
--
220 PSET(X,Y):A="X=?":GOSUB270
221 A=" " :GOSUB272:GOSUB280:X=VAL(C)
222 A=" " :GOSUB272:IFX>XLGOTO221
225 A="Y=?":GOSUB270
226 A=" " :GOSUB272:GOSUB280:Y=VAL(C)
227 A=" " :GOSUB272:IFY>YLGOTO226
229 A=" " :GOSUB272:IFY<0GOTO229
230 IFF=1 THENX1=X:Y1=Y:F=0:RETURN
231 IFF=0 THENX2=X:Y2=Y:F=1:RETURN
235 '-----TRACE D'UN CERCLE-----
--
240 A="R=?":GOSUB270
241 A=" " :GOSUB272:GOSUB280:R=VAL(C)
242 IFW<0ORW>127 THENA=" " :GOSUB272:GOT
0241
243 CIRCLE(X,Y),W
244 A=" " :GOSUB270:GOSUB272
245 RETURN
247 '-----AFFICHAGE D'UN CHR$-----
--
250 A="X=?":GOSUB270
251 A=" " :GOSUB272:GOSUB280:U=VAL(C)
252 A=" " :GOSUB272:IFU<0ORU>=HGOTO251
255 A="Y=?":GOSUB270
256 A=" " :GOSUB272:GOSUB280:V=VAL(C)
257 A=" " :GOSUB272:IFV<0ORV>=HGOTO256
259 A="C=?":GOSUB270
260 A=" " :GOSUB272:GOSUB280:W=VAL(C)
261 A=" " :GOSUB272:IFW<32ORW>255GOTO260
262 A=" " :GOSUB270
263 LOCATEU,V:PRINTCHR$(VAL(C));

```

Listing du programme (suite).


```

264 RETURN
267 '-----DIALOGUE-----
270 IFQ=0THENLOCATE17,0ELSELOCATES,3
271 GOTO275
272 IFQ=0THENLOCATE17,1ELSELOCATE10,3
273 GOTO275
274 LOCATE17,3
275 PRINTA;
276 IFQ=2ANDLEFT$(A,1)=" "ANDZ=1THENLINE
(0,24)-(6*L,24)
277 RETURN
278 '-----EQUIVALENT DE INPUT-----
280 J=STRIG(0):C=" "
281 D=INKEY$:IFD=" "THEN281
282 IFD=CHR$(13)THENRETURN
283 PRINTD:C=C+D:GOTO281
285 '
286 '*****FIN DU MODULE GRAPHIQUE*****
***
287 '
300 PSET(X,Y):GOSUB400
301 CLS:INPUT"Voulez-vous revoir le
dessin ";D
302 IFLEFT$(D,1)="O"THENGOSUB500
310 CLS:PRINT"Voulez-vous editer les FO
NT$ des caracteres graphiques en";
311 LOCATE0,3:INPUT"notation HEXA";D
312 IFLEFT$(D,1)<>"O"THEN330
320 CLS:INPUT" SORTIE EN CONTINU";D
322 IFLEFT$(D,1)="O"THENW=1ELSEW=0
325 GOSUB360
330 CLS:INPUT" Voulez-vous reprod
uire le dessin sur papier";D
331 IFLEFT$(D,1)="O"THENGOSUB540
335 IF=0GOTO340
336 CLS:INPUT"Voulez-vous verifier l'enr
egistrement";D
337 IFLEFT$(D,1)="O"THENCONSOLE0,1:GOSUB
600
340 CLS:INPUT" Voulez-vous initia-liser
les caracteresgraphiques";D
341 IFLEFT$(D,1)="O"THENCONSOLE0,1
350 CLS:INPUT"Voulez-vous faire unautre
dessin";D
352 CONSOLE,,1,1:IFLEFT$(D,1)="O"GOTO50
355 CLS:END
356 '---SORTIE DES FONT$ EN HEXADECIMAL-
---
360 LPRINTCT
365 S=R
370 FORI=0TOH-1:FORJ=0TOL-1
371 CLS:PRINT"FONT$( ";S;")=";FONT$(S)
372 LPRINT"FONT$( ";S;")=";FONT$(S)
373 IFW=0THENGOSUB70
375 S=S+1:IFS>159ANDS<224THENS=224
376 NEXTJ,I
377 RETURN
380 '---MODULE DE SAUVEGARDE SUR CASSETTE
---
400 IF=0GOTO420
402 A="CAS":GOSUB270
403 A="OK?":GOSUB272
404 BEEP0,3:BEEP0,1
405 A=" ":GOSUB272
406 BEEP0,3:BEEP0,1
407 IFINKEY$=""THEN403
408 J=STRIG(0):GOSUB270
409 '-----BLANC SONORE DE SEPARATION-----
410 MOTORON:BEEP0,60:BEEP0,1
411 INIT#1,"CASO:"
415 PRINT#1,CT,L,H,N,R
420 A="NUM":GOSUB270
421 S=R:F=0:K=0:D=CHR$(34)
425 IFIP=1THENLPRINTCT:LPRINT
427 '-----DEBUT BALAYAGE ECRAN-----
---
430 FORY=0TOH-1:FORX=0TOL-1
431 K=K+1
432 '---AFFICHAGE DU NUMERO DU CARACTERE-
---
433 A=RIGHT$(STR$(S),3):GOSUB272
434 GOSUB460:IFE=1THENGOSUB440
435 NEXTX,Y:RETURN
437 '-----FIN BALAYAGE-----
---
440 IF(LEN(D)+LEN(C)>253)GOTO445
441 D=D+C
442 IFK<NTHENRETURNELSEF=1
445 INIT#1,"CASO:"
446 D=D+CHR$(34)
447 PRINT#1,F,D
450 D=CHR$(34)+C
451 IFK=NANDF=0THEN442ELSERETURN
455 '-----CALCUL DU FONT$(S)-----
---
460 C=" ":FORI=Y*8TOY*8+7
461 U=0:V=0
462 FORJ=X*6TOX*6+5
463 IFPOINT(J,I)THENV=U+2^(7-U)
464 U=U+1:NEXTJ
465 A=STR$(V)
466 C=C+RIGHT$(A,LEN(A)-1)+","
467 NEXTI
468 BEEP10,2:BEEP0,1
469 '-----MEMORISATION-----
---
470 A=LEFT$(C,LEN(C)-1)
471 FONT$(S)=A
473 '---IMPRESSION DES PARAMETRES DECIMAU
X---
475 IFIP=1THENLPRINT"FONT$( ";S;")=";CHR$(
34);A;CHR$(34)
476 S=S+1:IFS>159ANDS<224THENS=224
477 RETURN
490 '-----AFFICHAGE D'UN DESSIN-----
---
500 CLS:S=R:FORY=0TOH-1
501 FORX=0TOL-1
502 LOCATEX,Y:PRINTCHR$(S);
503 S=S+1:IFS>159ANDS<224THENS=224
505 NEXTX,Y
506 GOSUB70:RETURN
520 '-----DESSIN SUR IMPRIMANTE X-710----
---
540 CLS:PRINT"L'imprimante est en servic
e ? ; couleur selectionnee ? ";
541 LOCATE3,3:PRINT"Alors O.K. ?";
542 GOSUB70
545 '-----TRACE DU CADRE-----
---
550 XD=64*H:X=240-XD/2:XI=-XD
551 YD=36*L-1:YI=-YD
552 LPRINTCHR$(17):LPRINTCHR$(18)
553 LPRINT"1"+STR$(X)+","+STR$(YI)
554 LPRINT"1"
560 LPRINT"J"+STR$(XD)+","0,0,"+STR$(YD)+
","+STR$(XI)+","0,0,"+STR$(YI)
561 LPRINT"S0"
562 YD=YD-1
565 GOSUB500
567 '-----RECOPIE POINT PAR POINT-----
---
570 FORI=0TOYL
571 IF(IMOD2)=1GOTO585

```

Listing du programme (suite).

Y = ? est alors affiché et, à nouveau, il faut entrer y, suivi de RETURN.

De la même façon une valeur, $y \geq H * 8$, n'est pas acceptée.

● Pour tracer une droite entre deux points A (x_1, y_1) et B (x_2, y_2), deux possibilités sont offertes :

- Tout d'abord, le **tracé à « vue »** où l'on ne connaîtra pas les coordonnées de A et B. Il faut amener le spot sur A, presser F6 (les coordonnées (x_1, y_1) sont alors mémorisées). On déplace ensuite le curseur en mode « effacement » jusqu'à B et on appuie à nouveau sur F6, puis sur la touche **D**.

- Ensuite, le **tracé précis**, pour lequel on utilise « l'option » **P** pour introduire successivement (x_1, y_1), puis (x_2, y_2). L'appui de **D** provoque le tracé de la droite désirée.

● Pour tracer un cercle de centre (x, y) et de rayon R, il est nécessaire de positionner d'abord le curseur au point (x, y), puis presser la touche **C**.

« R = ? » apparaît.

Il faut alors introduire le rayon suivi de RETURN.

● Pour « incruster » un caractère ASCII (code ≥ 32), il faut presser la touche **L**.

« X = ? » est affiché.

On introduit alors la colonne X ($0 \leq X \leq L - 1$) suivi de RETURN.

« Y = ? » est ensuite affiché.

Il faut alors introduire la ligne Y ($0 \leq Y \leq H - 1$) suivi de RETURN.

Enfin, « C = ? » est affiché.

Le code ASCII du caractère doit être fourni, suivi d'un RETURN.

● Pour effacer tout l'écran, il faut presser la touche **E**.

● Pour changer de dessin ou arrêter, il suffit de presser la touche **A**.

● Une fois le dessin achevé, une pression de la touche **M** lance la mémorisation.

Mémorisation et archivage

Dès que M a été pressée, si l'option sauvegarde a été demandée auparavant, le programme affiche un « CAS », suivi d'un « OK ? ».

Ceci pour rappeler aux étourdis de mettre le magnétophone en position « enregistrement ».

Une pression sur une touche quelconque lance l'exécution. Les « FONT\$ » sont calculées


```

572 XD=8*(I+1)
575 LPRINT"Q3":LPRINT"M"+STR$(XD)+",0"
580 FORJ=0TOXL
581 IFPOINT(J,I)THENLPRINT"P"ELSELPRINT
"P "
582 NEXTJ:GOTO595
585 XD=8*I
586 LPRINT"Q1":LPRINT"M"+STR$(XD)+", "+STR
R$(YD)
590 FORJ=XLTO0STEP-1
591 IFPOINT(J,I)THENLPRINT"P"ELSELPRINT
"P "
592 NEXTJ
595 NEXTI:RETURN

```

596 '-----RELECTURE D'UN ENREGISTREMENT-----

```

600 CLS:PRINT" Bande positionnee ? PLAY
? Volume et tonalite = 7/10 ?";
601 LOCATE1,3:PRINT" Alors allez-y ?";
602 GOSUB70
610 INIT#1,"CASI:"
612 INPUT#1,CT,L,H,N,R
613 CLS:PRINTCT,L;H;N;R
617 '-----LECTURE TANT QUE F=1-----
620 S=R
621 INPUT#1,F,C
622 A="":I=1
625 FORJ=1TOLEN(C)-1
626 D=MID$(C,J,1):IFD<>CHR$(44)THENA=A+D
:GOTO629
627 Q(I)=VAL(A):A=""
628 I=I+1:IFI>8THENGOSUB640
629 NEXTJ:Q(8)=VAL(A):GOSUB640
630 IFF=0GOTO621
631 'F=1 signale la fin d'enregistrement
635 GOSUB500:RETURN
637 '-----MEMORISATION-----
640 FONT$(S)="Q(1),Q(2),Q(3),Q(4),Q(5),Q
(6),Q(7),Q(8)"
642 CLS:PRINT"FONT$( ";S; )="";
643 FORP=1TO7:PRINTQ(P);",";NEXT
644 PRINTQ(8);
650 S=S+1:IFS>159ANDS<224THENS=224
651 A="":I=1:RETURN

```

690 '-----EXPLICATIONS-----

```

700 FORI=1TO18
702 READA:CLS:PRINTA;
704 GOSUB70:NEXT:RETURN
710 DATA" Avec un ecran fixe,PRESSEZ sur
une tou-che QUELCONQUE pour continuer"
715 DATA" Pour dessiner,utiliser les 4 t
ouches de deplacement du cur-seur"
720 DATA"Pour tracer une dia-gonale, app
uyer sur les 2 touches appropriees"
725 DATA"Pour passer en MODE LENT, pres
sez la barre d'espacement"
730 DATA"Dans ce cas,les coordonnees X,Y
du point s'affichent et cli-gnotent"
735 DATA"Pour revenir au MODERAPIDE,pres
sez a nouveau la barre"
740 DATA"Pour EFFACER,pressezd'abord la
touche defonction F6 situee au centre"
745 DATA" Dans ce cas une * est affiche
e en bas,a droite de l'ecran"
750 DATA"Pour revenir en MODE ECRITURE,p
ressez a nouveau la touche F6"
755 DATA"Pour positionner un point, pres
sez P,et introduisez ses coordonnees"
760 DATA"Pour tracer une droite entre 2
points A et B,pressez D..."
765 DATA"...apres avoir introduit les co

```

ordonnees de A et B par l'op-tion P"
770 DATA" Pour tracer un CERCLE, pre
ssez C etintroduisez le rayon"
775 DATA" Pour inscrire un caractere,p
ressez L et introduisez ... "
780 DATA"... la position (X,Y) et le code
ASCII du caractere"
785 DATA" Pour effacer TOUT l'ecran,pr
essez E"
790 DATA" Pour interrompre , pressez
A"
795 DATA" Une fois le dessin acheve , pr
essez M pour lancer la memo-risation"

Listing du programme (suite et fin).

**MICRO
SYSTEMES**

MICRO-SYSTEMES

```

FONT$( 128 )="0,00,124,124,124,120,120,120"
FONT$( 129 )="0,100,100,100,100,40,244,244"
FONT$( 130 )="0,70,100,100,100,100,230,230"
FONT$( 131 )="0,204,220,220,220,220,220,220"
FONT$( 132 )="0,220,220,220,220,150,120,120"
FONT$( 133 )="0,50,104,104,104,104,00,00"
FONT$( 134 )="0,240,252,252,252,20,20,124"
FONT$( 135 )="0,00,124,124,124,120,120,120"
FONT$( 136 )="0,104,100,100,100,00,00,00"
FONT$( 137 )="0,0,0,0,0,0,0,0"
FONT$( 138 )="0,0,0,0,0,0,0,0"
FONT$( 139 )="0,0,0,0,0,0,0,0"
FONT$( 140 )="0,0,0,0,0,0,0,0"
FONT$( 141 )="0,0,0,0,0,0,0,0"
FONT$( 142 )="0,0,0,0,0,0,0,0"
FONT$( 143 )="0,0,0,0,0,0,0,0"
FONT$( 144 )="120,120,120,120,120,120,120,0"
FONT$( 145 )="244,244,244,244,244,244,244,0"
FONT$( 146 )="230,230,230,230,230,230,230,0"
FONT$( 147 )="220,220,220,220,220,220,204,0"
FONT$( 148 )="120,150,150,100,100,100,100,0"
FONT$( 149 )="00,100,100,100,100,100,00,0"
FONT$( 150 )="120,120,124,00,00,00,20,0"
FONT$( 151 )="120,120,120,120,120,120,50,152,0"
FONT$( 152 )="00,00,00,252,252,252,240,0"
FONT$( 153 )="0,0,0,0,0,0,0,0"
FONT$( 154 )="0,0,0,0,0,0,0,0"
FONT$( 155 )="0,0,0,0,0,0,0,0"
FONT$( 156 )="0,0,0,0,0,0,0,0"
FONT$( 157 )="0,0,0,0,0,0,0,0"
FONT$( 158 )="0,0,0,0,0,0,0,0"
FONT$( 159 )="0,0,0,0,0,0,0,0"
FONT$( 224 )="0,00,124,124,120,120,124,124"
FONT$( 225 )="0,104,100,100,00,0,240,252"
FONT$( 226 )="0,120,120,120,120,120,124,124"
FONT$( 227 )="0,120,120,120,120,50,152,200"
FONT$( 228 )="0,124,252,252,240,240,252,252"
FONT$( 229 )="0,112,120,120,120,0,240,240"
FONT$( 230 )="0,252,252,252,252,0,00,00"
FONT$( 231 )="0,220,220,220,220,20,20,20"
FONT$( 232 )="0,124,124,124,00,120,120,120"
FONT$( 233 )="0,00,124,124,124,120,120,120"
FONT$( 234 )="0,100,100,100,100,40,244,244"
FONT$( 235 )="0,70,100,100,100,100,230,230"
FONT$( 236 )="0,100,100,100,150,132,132,252"
FONT$( 237 )="0,150,100,100,100,00,00,100"
FONT$( 238 )="0,220,220,220,20,0,252,252"
FONT$( 239 )="0,0,120,120,120,0,0,120"
FONT$( 240 )="00,0,120,120,120,120,50,0"
FONT$( 241 )="252,00,00,00,252,252,240,0"
FONT$( 242 )="124,124,00,20,12,12,12,0"
FONT$( 243 )="224,240,224,152,152,152,152,0"
FONT$( 244 )="124,0,240,240,244,244,116,0"
FONT$( 245 )="240,120,120,240,240,240,240,0"
FONT$( 246 )="00,00,00,00,00,00,00,0"
FONT$( 247 )="20,20,20,20,20,20,20,0"
FONT$( 248 )="252,252,120,120,100,100,100,0"
FONT$( 249 )="120,120,120,120,120,120,120,0"
FONT$( 250 )="244,244,244,244,244,244,244,0"
FONT$( 251 )="230,230,230,230,230,230,230,0"
FONT$( 252 )="252,252,152,152,220,220,220,0"
FONT$( 253 )="150,120,00,00,100,100,150,0"
FONT$( 254 )="252,20,20,20,124,124,124,0"
FONT$( 255 )="120,120,120,120,120,120,0,0"

```

Fig. 2. - Le Logo de Micro-Systèmes tel qu'il est imprimé. On constate que les points ne sont pas carrés comme sur l'écran du X 07.

une à une et éditées éventuellement sur l'imprimante. La sauvegarde s'effectue par « paquets » sur la cassette.

Chaque « FONT\$ » mémorisée entraîne l'affichage de son numéro sur l'écran. Un petit bip est émis.

Les questions auxquelles il a fallu répondre

● Comment créer un INPUT sans « dévastation » de l'écran ?

Le dialogue au cours des « options » P, C, L nécessite l'équivalent de la fonction INPUT classique. Or, il est impératif d'éviter tout déroulement dévastateur de l'écran. Le sous-programme des lignes 280 à 283 remplit cet office.

● Comment effectuer une recopie point par point « rapide » ?

L'idéal serait de disposer d'un pavé « carré » programmé dans les caractères de l'imprimante X-710. Malheureusement, ce n'est pas le cas.

A défaut, on utilise le caractère CHR\$(127) qui imprime un X (fig. 2) sur le papier. Son symbole sur l'écran est le fameux point d'interrogation à l'envers (¿) qui n'est accessible par aucune touche. Pour l'introduire dans les lignes 581 et 591, il faut procéder de la façon suivante :

Faire CLS, puis
 ?CHR\$(127) et RETURN

Le symbole ¿ est produit et il suffit de le déplacer sur l'écran pour l'introduire dans le texte des lignes 581 et 591 à l'endroit adéquat.

L'inconvénient (mineur) de l'utilisation de ce caractère est une légère déformation dans le sens de la hauteur de l'image imprimante par rapport au dessin présenté sur l'écran. Par contre, la recopie qui se fait dans les deux sens est relativement rapide.

● Comment compacter les enregistrements sur cassette ?

L'examen du module « sauvegarde » permet de voir que plusieurs « FONT\$ » successives sont « accrochées » les unes aux autres pour obtenir une variable D\$ d'une longueur aussi proche que possible de la longueur maximum autorisée pour une chaîne de caractères, soit 255 caractères.

Grâce à cette astuce, on limite le nombre d'enregistrements et on diminue considérablement le temps de sauvegarde (où, à l'inverse, de lecture d'un enregistrement). ■

Tableau des principales variables utilisées

Variables	Valeurs possibles	Rôle
A\$	quelconque	Affichage dans zone de dialogue
CT\$	quelconque	Titre de l'enregistrement
E	0	Pas d'enregistrement sur cassette
	1	Enregistrement sur cassette
F	0	Vaut zéro pour tout enregistrement d'une chaîne de FONT\$, sauf la dernière
	1	Signale la fin de l'enregistrement
H	$1 \leq H \leq 4$	Hauteur du cadre (nombre de lignes)
IP	0	Pas de sortie des FONT\$ sur imprimante
	1	Sortie sur imprimante
L	$1 \leq L \leq 20$	Largeur du cadre (nombre de colonnes)
N	1 à 64	Surface du dessin ($N = H * L$)
P	0	Mode effacement
	1	Mode dessin
Q		Localisation de la zone de dialogue :
	02	Utilisation des colonnes 17, 18, 19
	1	Utilisation de la ligne 3
	2	Restauration du trait du cadre lorsque $Q = 1$ et $H = 3$
R	128 à 159 224 à 255	Numéro du premier caractère graphique qui sera utilisé en mémorisation
S	idem R	Numéro du caractère courant en phase de mémorisation
T	0 à 8	Etat des touches de déplacement du curseur
X, Y	$0 \leq X \leq L$ $0 \leq Y \leq YL$	Coordonnées du spot
XL	6 à 120	Limite de x
YL	8 à 32	Limite de y
Z	1	Vitesse rapide
		Vitesse lente

Tableau des variables.

Particularités du programme

Le listing (fig. 3) est commenté et permet de repérer facilement la structure du programme :

Lignes 20 à 115 : module d'initialisation.

Lignes 120 à 283 : module graphique. C'est le « cœur » du programme.

Lignes 300 à 355 : traitements du dessin.

Lignes 400 à 477 : module de sauvegarde sur cassette et de mémorisation des FONT\$.

Lignes 500 à 506 : module d'affichage.

Lignes 540 à 595 : module de recopie point par point sur l'imprimante X-710.

Lignes 600 à 651 : module de relecture d'un enregistrement. Ce module est à recopier dans tout programme qui utilise un stockage de dessins sur cassette.

Lignes 700 à 795 : explications des fonctions offertes à l'utilisateur.

SUPER VEGAS MONTÉ

EXCEPTIONNEL

UNITE CENTRALE:

- microprocesseur 6809
- 64 Ko de mémoire R.A.M.
- 1 lecteur de disquettes double face (320Ko) possibilité d'extension à 4 lecteurs
- interfaces parallèles pour imprimante (type "Centronics") et manettes
- 2 E/S série RS 232C
- horloge temps réel avec batterie de sauvegarde
- sorties vidéo monochrome et couleur (RVB)

CLAVIER:

- clavier ergonomique 101 touches (détachable)
- pavés machine à écrire (Azerty), numérique et de fonctions
- adaptation parfaite au traitement de textes

LOGICIEL DE BASE: Flex et S. Basic

OPTIONS:

- carte graphique THR 512 x 512 } 8 couleurs
- carte graphique HR 256 x 256 } 2 plans
- boîtier SS30 pour connexions des cartes :
 - interface S.A.S.I. (SCSI)
 - interface IEEE 488
 - digitalisation d'images
 - programmeur d'EPROMS
 - convertisseurs AD et DA
 - synthèse vocale
 - ANTIOPE
- etc..



LANGAGES ET LOGICIELS :

- langage C, PL 9, Pascal, Forth, assembleurs
- nombreux outils de développement et dessin assisté.

Monté en coffret noir métal, connecteurs très haute fiabilité (type militaire). Prix:

14 950 F TTC

Offre valable dans la limite des stocks disponibles

Vegas

Circuit imprimé + 2 Eproms

+ 1 disque Flex + dossier de montage _ 1 200 F

Clavier

Azerty 101 touches matricé X,Y _____ 590 F

Codeur en kit interface // _____ 350 F

Coffret _____ 290 F



15, Quai Jules Guesde 94400 Vitry - Tél. (1) 681.88.37